

NAG Library Routine Document

F12FGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

Note: *this routine uses optional parameters to define choices in the problem specification. If you wish to use default settings for all of the optional parameters, then the option setting routine F12FDF need not be called. If, however, you wish to reset some or all of the settings please refer to Section 10 in F12FDF for a detailed description of the specification of the optional parameters.*

1 Purpose

F12FGF is the main solver routine in a suite of routines consisting of F12FDF, F12FFF and F12FGF, that must be called following an initial call to F12FFF and following any calls to F12FDF.

F12FGF returns approximations to selected eigenvalues, and (optionally) the corresponding eigenvectors, of a standard or generalized eigenvalue problem defined by real banded symmetric matrices. The banded matrix must be stored using the LAPACK storage format for real banded nonsymmetric matrices.

2 Specification

```

SUBROUTINE F12FGF(KL, KU, AB, LDAB, MB, LDMB, SIGMA, NCONV, D, Z, LDZ,
1             RESID, V, LDV, COMM, ICOMM, IFAIL)
      INTEGER          KL, KU, LDAB, LDMB, NCONV, LDZ, LDV, ICOMM(*), IFAIL
      double precision AB(LDAB,*), MB(LDMB,*), SIGMA, D(*), Z(LDZ,*),
1             RESID(*), V(LDV,*), COMM(*)

```

3 Description

The suite of routines is designed to calculate some of the eigenvalues, λ , (and optionally the corresponding eigenvectors, x) of a standard eigenvalue problem $Ax = \lambda x$, or of a generalized eigenvalue problem $Ax = \lambda Bx$ of order n , where n is large and the coefficient matrices A and B are banded, real and symmetric.

Following a call to the initialization routine F12FFF, F12FGF returns the converged approximations to eigenvalues and (optionally) the corresponding approximate eigenvectors and/or an orthonormal basis for the associated approximate invariant subspace. The eigenvalues (and eigenvectors) are selected from those of a standard or generalized eigenvalue problem defined by real banded symmetric matrices. There is negligible additional computational cost to obtain eigenvectors; an orthonormal basis is always computed, but there is an additional storage cost if both are requested.

The banded matrices A and B must be stored using the LAPACK storage format for banded nonsymmetric matrices; please refer to Section 3.3.2 in the F07 Chapter Introduction for details on this storage format.

F12FGF is based on the banded driver routines **dsbdr1** to **dsbdr6** from the ARPACK package, which uses the Implicitly Restarted Lanczos iteration method. The method is described in Lehoucq and Sorensen (1996) and Lehoucq (2001) while its use within the ARPACK software is described in great detail in Lehoucq *et al.* (1998). This suite of routines offers the same functionality as the ARPACK banded driver software for real symmetric problems, but the interface design is quite different in order to make the option setting clearer and to combine the different drivers into a general purpose routine.

F12FGF, is a general purpose forward communication routine that must be called following initialization by F12FFF. F12FGF uses options, set either by default or explicitly by calling F12FDF, to return the converged approximations to selected eigenvalues and (optionally):

- the corresponding approximate eigenvectors;
- an orthonormal basis for the associated approximate invariant subspace;
- both.

4 References

Lehoucq R B (2001) Implicitly Restarted Arnoldi Methods and Subspace Iteration *SIAM Journal on Matrix Analysis and Applications* **23** 551–562

Lehoucq R B and Scott J A (1996) An evaluation of software for computing eigenvalues of sparse nonsymmetric matrices *Preprint MCS-P547-1195* Argonne National Laboratory

Lehoucq R B and Sorensen D C (1996) Deflation Techniques for an Implicitly Restarted Arnoldi Iteration *SIAM Journal on Matrix Analysis and Applications* **17** 789–821

Lehoucq R B, Sorensen D C and Yang C (1998) *ARPACK Users' Guide: Solution of Large-Scale Eigenvalue Problems with Implicitly Restarted Arnoldi Methods* SIAM, Philadelphia

5 Parameters

- 1: KL – INTEGER *Input*
On entry: the number of subdiagonals of the matrices *A* and *B*.
Constraint: $KL \geq 0$.
- 2: KU – INTEGER *Input*
On entry: the number of superdiagonals of the matrices *A* and *B*. Since *A* and *B* are symmetric, the normal case is $KU = KL$.
Constraint: $KU \geq 0$.
- 3: AB(LDAB,*) – **double precision** array *Input*
Note: the second dimension of the array AB must be at least $\max(1, N)$ (see F12FFF).
On entry: must contain the matrix *A* in LAPACK banded storage format for nonsymmetric matrices (see Section 3.3.2 in the F07 Chapter Introduction).
- 4: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the (sub)program from which F12FGF is called.
Constraint: $LDAB \geq 2 \times KL + KU + 1$.
- 5: MB(LDMB,*) – **double precision** array *Input*
Note: the second dimension of the array MB must be at least $\max(1, N)$ (see F12FFF).
On entry: must contain the matrix *B* in LAPACK banded storage format for nonsymmetric matrices (see Section 3.3.2 in the F07 Chapter Introduction).
- 6: LDMB – INTEGER *Input*
On entry: the first dimension of the array MB as declared in the (sub)program from which F12FGF is called.
Constraint: $LDMB \geq 2 \times KL + KU + 1$.
- 7: SIGMA – **double precision** *Input*
On entry: if one of the Shifted Inverse modes has been selected then SIGMA contains the real shift used; otherwise SIGMA is not referenced.

- 8: NCONV – INTEGER *Output*
On exit: the number of converged eigenvalues.
- 9: D(*) – **double precision** array *Output*
Note: the dimension of the array D must be at least NCV (see F12FFF).
On exit: the first NCONV locations of the array D contain the converged approximate eigenvalues.
- 10: Z(LDZ,*) – **double precision** array *Output*
Note: the second dimension of the array Z must be at least NCV + 1 if the default option Vectors = Ritz has been selected and at least 1 if the option Vectors = None or Schur has been selected (see F12FFF).
On exit: if the default option Vectors = Ritz has been selected then Z contains the final set of eigenvectors corresponding to the eigenvalues held in D. The real eigenvector associated with eigenvalue i , for $i = 0, 1, \dots, \text{NCONV} - 1$, is stored in the i th column of Z.
- 11: LDZ – INTEGER *Input*
On entry: the first dimension of the array Z as declared in the (sub)program from which F12FGF is called.
Constraints:
 if the default option Vectors = Ritz has been selected, $\text{LDZ} \geq \text{N}$;
 if the option Vectors = None or Schur has been selected, $\text{LDZ} \geq 1$.
- 12: RESID(*) – **double precision** array *Input/Output*
Note: the dimension of the array RESID must be at least N (see F12FFF).
On entry: need not be set unless the option Initial Residual has been set in a prior call to F12FDF in which case RESID should contain an initial residual vector.
On exit: contains the final residual vector.
- 13: V(LDV,*) – **double precision** array *Output*
Note: the second dimension of the array V must be at least $\max(1, \text{NCV})$ (see F12FFF).
On exit: if the option Vectors has been set to Schur or Ritz and a separate array Z has been passed then the first $\text{NCONV} \times n$ elements of V will contain approximate Schur vectors that span the desired invariant subspace. The i th Schur vector is stored in the i th column of V.
- 14: LDV – INTEGER *Input*
On entry: the first dimension of the array V as declared in the (sub)program from which F12FGF is called.
Constraint: $\text{LDV} \geq n$.
- 15: COMM(*) – **double precision** array *Communication Array*
Note: the dimension of the array COMM must be at least $\max(1, \text{LCOMM})$ (see F12FFF).
On initial entry: must remain unchanged from the prior call to F12FDF and F12FFF.
On exit: contains no useful information.
- 16: ICOMM(*) – INTEGER array *Communication Array*
Note: the dimension of the array ICOMM must be at least $\max(1, \text{LICOMM})$ (see F12FFF).
On initial entry: must remain unchanged from the prior call to F12FDF and F12FDF.
On exit: contains no useful information.

17: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $KL < 0$.

IFAIL = 2

On entry, $KU < 0$.

IFAIL = 3

On entry, $LDAB < 2 \times KL + KU + 1$.

IFAIL = 4

Iteration Limit < 0 .

IFAIL = 5

The options Generalized and Regular are incompatible.

IFAIL = 6

Eigenvalues from Both Ends of the spectrum were requested, but only one eigenvalue (NEV) is requested.

IFAIL = 7

The Initial Residual was selected but the starting vector held in RESID is zero.

IFAIL = 8

On entry, $LDZ < \max(1, N)$ or $LDZ < 1$ when no vectors are required.

IFAIL = 9

On entry, the option Vectors = Select was selected, but this is not yet implemented.

IFAIL = 10

The number of eigenvalues found to sufficient accuracy is zero.

IFAIL = 11

Could not build a Lanczos factorization. Consider changing NCV or NEV in the initialization routine (see Section 5 in F12FAF for details of these parameters).

IFAIL = 12

Unexpected error in internal call to compute eigenvalues and corresponding error bounds of the current symmetric tridiagonal matrix. Please contact NAG.

IFAIL = 13

Unexpected error during calculation of a real Schur form: there was a failure to compute all the converged eigenvalues. Please contact NAG.

IFAIL = 14

Failure during internal factorization of real banded matrix. Please contact NAG.

IFAIL = 15

Failure during internal solution of real banded system. Please contact NAG.

IFAIL = 16

The maximum number of iterations has been reached. Some Ritz values may have converged; NCONV returns the number of converged values.

IFAIL = 17

No shifts could be applied during a cycle of the implicitly restarted Lanczos iteration. One possibility is to increase the size of NCV relative to NEV (see Section 5 in F12FFF for details of these parameters).

IFAIL = 18

An unexpected error has occurred. Please contact NAG.

IFAIL = 19

The routine was unable to dynamically allocate sufficient internal workspace. Please contact NAG.

7 Accuracy

The relative accuracy of a Ritz value, λ , is considered acceptable if its Ritz estimate $\leq \text{Tolerance} \times |\lambda|$. The default Tolerance used is the *machine precision* given by X02AJF.

8 Further Comments

None.

9 Example

This example solves $Ax = \lambda x$ in regular mode, where A is obtained from the standard central difference discretization of the two-dimensional convection-diffusion operator $\frac{d^2 u}{dx^2} + \frac{d^2 u}{dy^2} = \rho \frac{du}{dx}$ on the unit square with zero Dirichlet boundary conditions. A is stored in LAPACK banded storage format.

9.1 Program Text

```
*      F12FGF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          LICOMM, NIN, NOUT
      PARAMETER        (LICOMM=140, NIN=5, NOUT=6)
      INTEGER          MAXBDW, MAXN, MAXNCV, LDAB, LDV
      PARAMETER        (MAXBDW=50, MAXN=1000, MAXNCV=50, LDAB=MAXBDW,
+                     LDV=MAXN)
      INTEGER          LCOMM
```

```

PARAMETER          (LCOMM=60)
DOUBLE PRECISION ONE, ZERO
PARAMETER          (ONE=1.0D+0,ZERO=0.0D+0)
*
.. Local Scalars ..
DOUBLE PRECISION H2, SIGMA
INTEGER            I, IDIAG, IFAIL, IFAIL1, ISUB, ISUP, J, KL, KU,
+
LO, N, NCONV, NCV, NEV, NX, OUTCHN
*
.. Local Arrays ..
DOUBLE PRECISION AB(LDAB,MAXN), AX(MAXN), COMM(LCOMM),
+
D(MAXNCV,2), MB(1), RESID(MAXN), V(LDV,MAXNCV)
INTEGER            ICOMM(LICOMM)
*
.. External Functions ..
DOUBLE PRECISION DNRM2
EXTERNAL           DNRM2
*
.. External Subroutines ..
EXTERNAL           DAXPY, DGBMV, F06QHF, F12FFF, F12FGF, X04ABF,
+
X04CAF
*
.. Intrinsic Functions ..
INTRINSIC          DABS
*
.. Executable Statements ..
WRITE (NOUT,*) 'F12FGF Example Program Results'
OUTCHN = NOUT
WRITE (NOUT,*)
*
Skip heading in data file
READ (NIN,*)
READ (NIN,*) NX, NEV, NCV
N = NX*NX
IF (N.LT.1 .OR. N.GT.MAXN) THEN
  WRITE (NOUT,99999) 'N is out of range: N = ', N
ELSE IF (NCV.GT.MAXNCV) THEN
  WRITE (NOUT,99999) 'NCV is out of range: NCV = ', NCV
ELSE
  IFAIL = 1
*
  Initialize communication arrays.
  CALL F12FFF(N,NEV,NCV,ICOMM,LICOMM,COMM,LCOMM,IFAIL)
*
  IF (IFAIL.EQ.0) THEN
*
*
*
  Construct the matrix A in banded form and store in AB.
  Zero out AB.
  IFAIL = 0
  CALL F06QHF('G',LDAB,N,ZERO,ZERO,AB,LDAB)
*
  KU, KL are number of superdiagonals and subdiagonals within
  the band of matrices A and M.
  KL = NX
  KU = NX
*
  Main diagonal of A.
  H2 = ONE/((NX+1)*(NX+1))
  IDIAG = KL + KU + 1
  DO 20 J = 1, N
    AB(IDIAG,J) = 4.0D+0/H2
20  CONTINUE
*
  First subdiagonal and superdiagonal of A.
  ISUP = KL + KU
  ISUB = KL + KU + 2
  DO 60 I = 1, NX
    LO = (I-1)*NX
    DO 40 J = LO + 1, LO + NX - 1
      AB(ISUP,J+1) = -ONE/H2
      AB(ISUB,J) = -ONE/H2
40  CONTINUE
60  CONTINUE
*
  KL-th subdiagonal and KU-th super-diagonal.
  ISUP = KL + 1
  ISUB = 2*KL + KU + 1
  DO 100 I = 1, NX - 1
    LO = (I-1)*NX
    DO 80 J = LO + 1, LO + NX
      AB(ISUP,NX+J) = -ONE/H2
      AB(ISUB,J) = -ONE/H2
80  CONTINUE

```

```

100      CONTINUE
*
*      Find eigenvalues of largest magnitude and the corresponding
*      eigenvectors.
      IFAIL = -1
      CALL F12FGF(KL,KU,AB,LDAB,MB,1,SIGMA,NCONV,D,V,LDV,RESID,V,
+           LDV,COMM,ICOMM,IFAIL)
      IF (IFAIL.EQ.0) THEN
*      Compute the residual norm ||A*x - lambda*x||.
      DO 120 J = 1, NCONV
          CALL DGBMV('NoTranspose',N,N,KL,KU,ONE,AB(KL+1,1),
+           LDAB,V(1,J),1,ZERO,AX,1)
          CALL DAXPY(N,-D(J,1),V(1,J),1,AX,1)
          D(J,2) = DNRM2(N,AX,1)
          D(J,2) = D(J,2)/DABS(D(J,1))
120      CONTINUE
          WRITE (NOUT,*)
          CALL XO4ABF(1,OUTCHN)
          CALL XO4CAF('G','N',NCONV,2,D,MAXNCV,
+           ' Ritz values and residuals',IFAIL1)
      END IF
      ELSE
          WRITE (NOUT,99998) IFAIL
      END IF
      END IF
*
99999  FORMAT (1X,A,I5)
99998  FORMAT (1X,' ** F12FFF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

F12FGF Example Program Data

10 4 10 : Values for NX NEV and NCV

9.3 Program Results

F12FGF Example Program Results

Ritz values and residuals		
	1	2
1	8.9117E+02	1.8474E-15
2	9.1978E+02	3.1522E-15
3	9.1978E+02	5.5943E-16
4	9.4839E+02	1.0295E-15
